

Using a data logger as a Remote Telemetry unit via SMS.

There are many applications where a *dataTaker* data logger is used as a Remote Telemetry Unit. Using the following technique we can easily and robustly collect data from multiple data loggers and provide calculated results, control, alarms and reports. When we also incorporate SMS reporting and SMS alarms we can have a powerful flexible and reliable system.

When used as an RTU, the *dataTaker* data logger is configured to return data when the Poll character, an X, is received.

Using the alarm function in conjunction with the *dataTaker* SMSX modem an SMS message can be sent to another *dataTaker* data logger where the results can be stored and / or used in other calculations.


This also allows for complex applications where a number of remote loggers are returning data to a single *dataTaker* data logger for calculation and data storage.

A possible application is temperature matched curing of concrete where temperature in a water bath in a laboratory is set to the temperature measured in the field. This is to ensure the concrete samples in the laboratory are exposed to exactly the same conditions as the concrete in the field.

Method:

Each logger is connected to a *dataTaker* SMSX modem any remote logger can send its readings to another logger via alarm statements in this form: 1CV(W)=22.5

The central loggers will receive messages via SMS into its communications port in the form 1CV(W)=22.5 2CV(W)=1024 etc. A single *dataTaker* data logger can collect multiple readings from multiple data loggers. This data logger can provide the appropriate calculations, control, alarms and reporting functions.



Code

```
P36=0
$="Site1"
BEGIN
  RX
    1TK("Temp",=1CV)
    2CV(W)=1CV+1
    3CV(W)=1CV-1
    92CV(W)=1
  RZ30S
    ALARMR1(1TK<>3CV,2CV)1WARN"[1DSO=1 X]"
    ALARMR2(92CV><1,2/90S)AND
    ALARMR3(1CV<999999)2WARN"et-
sms=0|1CV(W)=?|0409174846|90^M[91CV(W)=91CV+1]"
    ALARMR4(92CV>2/30S)OR
    ALARMR5(91CV>5/30S)"[1DSO=0 92CV(W)=0 91CV(W)=0]"
  END
/O/q/r/e/m
LOGON
G
```

This is an example of code that measures temperature. The logger will check the temperature every 30 seconds. If the temperature is found to have changed from the last recorded reading by less than 1 degree no action is taken. If it is found to have varied by MORE than one degree, digital output one is set to turn power on to the modem and then the measurements are recorded using the X schedule and new thresholds are set and a working variable 92CV is set to 1. The alarm 2 and 3 are now true and 90 seconds after they are true the SMS message is sent. When the message has been received and sent by the SMSX modem it responds with a 92CV=92CV+1, this increments the 92CV to 2 otherwise if the message is NOT correctly sent (i.e 92CV is not incremented to zero) the SMS message will be resent. The number of times it is resent is counted using the 91CV variable. The SMS message is sent to the 'master' logger in the form 1CV(W)=? (The current reading is substituted for the question mark). 30 seconds after 92CV is set to 2 or after 5 attempts have been made to send the message alarm 4 turns the power off to the modem and sets 92CV back to zero and the message counter 91CV back to zero ready for another value to be transmitted. Using the alarm 1 structure the number of messages sent is minimized. In this case the 'master' logger could use 1CV as the set point for the control of temperature on a cooling bath, record it and/or use the 1CV value in a calculation.

Conclusion

SMS messaging provides some significant benefits over other communication possibilities. SMS messages do not require the signal quality that standard dialup or GPRS require; also the software overhead of dialup is significant. GPRS requires additional infrastructure because of IP addressing issues. So for simplicity and reliability the SMS messaging is a smart solution to this problem. It should be noted that the time lapse of messages sent via SMS is around 60 seconds.

Datataker Pty Ltd - 7 Seismic Court Rowville Victoria 3178 Australia
Tel: 03 9764 8600 +61 3 9764 8600 Fax: 03 9764 8997 +61 3 9764 8997
E-mail: sales@datataker.com.au Web: www.datataker.com